

NAG Library Routine Document

F08JVF (ZSTEDC)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08JVF (ZSTEDC) computes all the eigenvalues and, optionally, all the eigenvectors of a real n by n symmetric tridiagonal matrix, or of a complex full or banded Hermitian matrix which has been reduced to tridiagonal form.

2 Specification

```
SUBROUTINE F08JVF (COMPZ, N, D, E, Z, LDZ, WORK, LWORK, RWORK, LRWORK,      &
                  IWORK, LIWORK, INFO)
INTEGER                N, LDZ, LWORK, LRWORK, IWORK(max(1,LIWORK)),      &
                  LIWORK, INFO
REAL (KIND=nag_wp)    D(*), E(*), RWORK(max(1,LRWORK))
COMPLEX (KIND=nag_wp) Z(LDZ,*), WORK(max(1,LWORK))
CHARACTER(1)          COMPZ
```

The routine may be called by its LAPACK name ***zstedc***.

3 Description

F08JVF (ZSTEDC) computes all the eigenvalues and, optionally, the eigenvectors of a real symmetric tridiagonal matrix T . That is, the routine computes the spectral factorization of T given by

$$T = Z\Lambda Z^T,$$

where Λ is a diagonal matrix whose diagonal elements are the eigenvalues, λ_i , of T and Z is an orthogonal matrix whose columns are the eigenvectors, z_i , of T . Thus

$$Tz_i = \lambda_i z_i, \quad i = 1, 2, \dots, n.$$

The routine may also be used to compute all the eigenvalues and eigenvectors of a complex full, or banded, Hermitian matrix A which has been reduced to real tridiagonal form T as

$$A = QTQ^H,$$

where Q is unitary. The spectral factorization of A is then given by

$$A = (QZ)\Lambda(QZ)^H.$$

In this case Q must be formed explicitly and passed to F08JVF (ZSTEDC) in the array Z , and the routine called with $\text{COMPZ} = \text{'V'}$. Routines which may be called to form T and Q are

full matrix	F08FSF (ZHETRD) and F08FTF (ZUNGTR)
full matrix, packed storage	F08GSF (ZHPTRD) and F08GTF (ZUPGTR)
band matrix	F08HSF (ZHBTRD), with $\text{VECT} = \text{'V'}$

When only eigenvalues are required then this routine calls F08JFF (DSTERF) to compute the eigenvalues of the tridiagonal matrix T , but when eigenvectors of T are also required and the matrix is not too small, then a divide and conquer method is used, which can be much faster than F08JSF (ZSTEQR), although more storage is required.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: COMPZ – CHARACTER(1) *Input*
On entry: indicates whether the eigenvectors are to be computed.
 COMPZ = 'N'
 Only the eigenvalues are computed (and the array Z is not referenced).
 COMPZ = 'V'
 The eigenvalues and eigenvectors of A are computed (and the array Z must contain the matrix Q on entry).
 COMPZ = 'I'
 The eigenvalues and eigenvectors of T are computed (and the array Z is initialized by the routine).
Constraint: COMPZ = 'N', 'V' or 'I'.
- 2: N – INTEGER *Input*
On entry: n , the order of the symmetric tridiagonal matrix T .
Constraint: $N \geq 0$.
- 3: D(*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: the diagonal elements of the tridiagonal matrix.
On exit: if INFO = 0, the eigenvalues in ascending order.
- 4: E(*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array E must be at least $\max(1, N - 1)$.
On entry: the subdiagonal elements of the tridiagonal matrix.
On exit: E is overwritten.
- 5: Z(LDZ,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array Z must be at least $\max(1, N)$ if COMPZ = 'V' or 'I', and at least 1 otherwise.
On entry: if COMPZ = 'V', Z must contain the unitary matrix Q used in the reduction to tridiagonal form.
On exit: if COMPZ = 'V', Z contains the orthonormal eigenvectors of the original Hermitian matrix A , and if COMPZ = 'I', Z contains the orthonormal eigenvectors of the symmetric tridiagonal matrix T .
 If COMPZ = 'N', Z is not referenced.
- 6: LDZ – INTEGER *Input*
On entry: the first dimension of the array Z as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

Constraints:

if COMPZ = 'V' or 'T', $LDZ \geq \max(1, N)$;
otherwise $LDZ \geq 1$.

- 7: WORK(max(1, LWORK)) – COMPLEX (KIND=nag_wp) array *Workspace*

On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.

- 8: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK, RWORK and IWORK arrays, returns these values as the first entries of the WORK, RWORK and IWORK arrays, and no error message related to LWORK, LRWORK or LIWORK is issued.

Constraints: if LWORK $\neq -1$,

if COMPZ = 'N' or 'T' or $N \leq 1$, LWORK must be at least 1;
if COMPZ = 'V' and $N > 1$, LWORK must be at least N^2 .

Note: that for COMPZ = 'V', then if N is less than or equal to the minimum divide size, usually 25, then LWORK need only be 1.

- 9: RWORK(max(1, LRWORK)) – REAL (KIND=nag_wp) array *Workspace*

On exit: if INFO = 0, RWORK(1) returns the optimal LRWORK.

- 10: LRWORK – INTEGER *Input*

On entry: the dimension of the array RWORK as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

If LRWORK = -1, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK, RWORK and IWORK arrays, returns these values as the first entries of the WORK, RWORK and IWORK arrays, and no error message related to LWORK, LRWORK or LIWORK is issued.

Constraints: if LRWORK $\neq -1$,

if COMPZ = 'N' or $N \leq 1$, LRWORK must be at least 1;
if COMPZ = 'V' and $N > 1$,
LRWORK must be at least $1 + 3 \times N + 2 \times N \times \lg(N) + 4 \times N^2$, where $\lg(N)$ = smallest integer k such that $2^k \geq N$;
if COMPZ = 'T' and $N > 1$, LRWORK must be at least $1 + 4 \times N + 2 \times N^2$.

Note: that for COMPZ = 'V' or 'T' then if N is less than or equal to the minimum divide size, usually 25, then LRWORK need only be $\max(1, 2 \times (N - 1))$.

- 11: IWORK(max(1, LIWORK)) – INTEGER array *Workspace*

On exit: if INFO = 0, IWORK(1) returns the optimal LIWORK.

- 12: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08JVF (ZSTEDC) is called.

If LIWORK = -1, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK, RWORK and IWORK arrays, returns these values as the first entries of the WORK, RWORK and IWORK arrays, and no error message related to LWORK, LRWORK or LIWORK is issued.

Constraints: if LIWORK $\neq -1$,

if COMPZ = 'N' or $N \leq 1$, LIWORK must be at least 1;

if COMPZ = 'V' or $N > 1$, LIWORK must be at least $6 + 6 \times N + 5 \times N \times \lg(N)$;

if COMPZ = 'T' or $N > 1$, LIWORK must be at least $3 + 5 \times N$.

13: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm failed to compute an eigenvalue while working on the submatrix lying in rows and columns INFO/(N + 1) through INFO mod (N + 1).

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(T + E)$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the *machine precision*.

If λ_i is an exact eigenvalue and $\tilde{\lambda}_i$ is the corresponding computed value, then

$$|\tilde{\lambda}_i - \lambda_i| \leq c(n)\epsilon\|T\|_2,$$

where $c(n)$ is a modestly increasing function of n .

If z_i is the corresponding exact eigenvector, and \tilde{z}_i is the corresponding computed eigenvector, then the angle $\theta(\tilde{z}_i, z_i)$ between them is bounded as follows:

$$\theta(\tilde{z}_i, z_i) \leq \frac{c(n)\epsilon\|T\|_2}{\min_{i \neq j} |\lambda_i - \lambda_j|}.$$

Thus the accuracy of a computed eigenvector depends on the gap between its eigenvalue and all the other eigenvalues.

See Section 4.7 of Anderson *et al.* (1999) for further details. See also F08FLF (DDISNA).

8 Parallelism and Performance

F08JVF (ZSTEDC) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08JVF (ZSTEDC) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

If only eigenvalues are required, the total number of floating-point operations is approximately proportional to n^2 . When eigenvectors are required the number of operations is bounded above by approximately the same number of operations as F08JSF (ZSTEQR), but for large matrices F08JVF (ZSTEDC) is usually much faster.

The real analogue of this routine is F08JHF (DSTEDC).

10 Example

This example finds all the eigenvalues and eigenvectors of the Hermitian band matrix

$$A = \begin{pmatrix} -3.13 & 1.94 - 2.10i & -3.40 + 0.25i & 0 \\ 1.94 + 2.10i & -1.91 & -0.82 - 0.89i & -0.67 + 0.34i \\ -3.40 - 0.25i & -0.82 + 0.89i & -2.87 & -2.10 - 0.16i \\ 0 & -0.67 - 0.34i & -2.10 + 0.16i & 0.50 \end{pmatrix}.$$

A is first reduced to tridiagonal form by a call to F08HSF (ZHBTRD).

10.1 Program Text

```

Program f08jvfe

!      F08JVF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
      Use nag_library, Only: dznrm2, nag_wp, x04dbf, zhbtrd, zstedc
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter    :: uplo = 'U'
!      .. Local Scalars ..
      Complex (Kind=nag_wp)       :: scal
      Integer                     :: i, ifail, info, j, k, kd, ldab, ldz, &
                                   lgn, liwork, lrwork, lwork, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,,:), work(:,), z(:,,:)
      Complex (Kind=nag_wp)              :: cdum(1)
      Real (Kind=nag_wp), Allocatable    :: d(:,), e(:,), rwork(:)
      Real (Kind=nag_wp)                 :: rdum(1)
      Integer                             :: idum(1)
      Integer, Allocatable                :: iwork(:)
      Character (1)                       :: clabs(1), rlabs(1)
!      .. Intrinsic Procedures ..
      Intrinsic                          :: abs, ceiling, conjg, log, max,      &
                                   maxloc, min, nint, real
!      .. Executable Statements ..
      Write (nout,*) 'F08JVF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      ldz = n
      lgn = ceiling(log(real(n,kind=nag_wp))/log(2.0_nag_wp))
      Allocate (ab(ldab,n),z(ldz,n),d(n),e(n-1))

!      Use routine workspace query to get optimal workspace.
      lwork = -1
      lrwork = -1
      liwork = -1
!      The NAG name equivalent of zstedc is f08jvf
      Call zstedc('V',n,d,e,z,ldz,cdum,lwork,rdum,lrwork,idum,liwork,info)

```

```

!      Make sure that there is enough workspace.
      lwork = max(n*n,nint(real(cdum(1))))
      lrwork = max(1+3*n+2*n*lgm+4*n*n,nint(rdum(1)))
      liwork = max(6+6*n+5*n*lgm,idum(1))
      Allocate (work(lwork),rwork(lrwork),iwork(liwork))

!      Read the upper or lower triangular part of the band matrix A
!      from data file

      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If

!      Reduce A to tridiagonal form T = (Z**T)*A*Z, and form Z

!      The NAG name equivalent of zhbtrd is f08hsf
      Call zhbtrd('V',uplo,n,kd,ab,ldab,d,e,z,ldz,work,info)

!      Calculate all the eigenvalues and eigenvectors of A,
!      from T and Z

!      The NAG name equivalent of zstedc is f08jvf
      Call zstedc('V',n,d,e,z,ldz,work,lwork,rwork,lrwork,iwork,liwork,info)

      If (info==0) Then

!      Print eigenvalues and eigenvectors

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) d(1:n)

        Write (nout,*)
        Flush (nout)

!      Normalize the eigenvectors, largest element real
        Do i = 1, n
          rwork(1:n) = abs(z(1:n,i))
          k = maxloc(rwork(1:n),1)
          scal = conjg(z(k,i))/abs(z(k,i))/dznrm2(n,z(1,i),1)
          z(1:n,i) = z(1:n,i)*scal
        End Do

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf('General',' ',n,n,z,ldz,'Bracketed','F7.4','Eigenvectors', &
          'Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
        Write (nout,99998) 'Failure in ZSTEDC. INFO = ', info
      End If

99999 Format (4X,F8.4,3(10X,F8.4))
99998 Format (1X,A,I10)
      End Program f08jvfe

```

10.2 Program Data

F08JVF Example Program Data

```

      4              2                      :Values of N and KD
      ( -3.13 , 0.00) (  1.94, -2.10) ( -3.40,  0.25)
              ( -1.91,  0.00) ( -0.82, -0.89) ( -0.67,  0.34)
                      ( -2.87,  0.00) ( -2.10, -0.16)
                                (  0.50,  0.00) :End matrix A

```

10.3 Program Results

F08JVF Example Program Results

```

Eigenvalues
      -7.0042          -4.0038          0.5968          3.0012

```

```

Eigenvectors
              1              2              3              4
1  ( 0.7293, 0.0000) (-0.2128, 0.1511) (-0.3354,-0.1604) (-0.5114,-0.0163)
2  (-0.1654,-0.2046) ( 0.7316, 0.0000) (-0.2804,-0.3413) (-0.2374,-0.3796)
3  ( 0.6081, 0.0301) ( 0.3910,-0.3843) (-0.0144, 0.1532) ( 0.5523,-0.0000)
4  ( 0.1653,-0.0303) ( 0.2775,-0.1378) ( 0.8019, 0.0000) (-0.4517, 0.1693)

```
